Joins and secant varieties

Chris Peterson and Hirotachi Abo

Note. This script is also available at:

http://www.math.colostate.edu/~abo/Research/smi/smi-algebraic-geometry.html

There you can find problem sets and `Macaulay2` scripts as well.

1. The join of two varieties

Let k be an algebraically closed field, let $\mathbb{P}^n$ be the n-dimensional projective space over k, and let V and W be two disjoint irreducible projective varieties in $\mathbb{P}^n$. We denote by $J(V,W)$ the union of the lines in $\mathbb{P}^n$ joining V to W. Then $J(V,W)$ is a projective variety (see pages 69 and 70 in *Algebraic Geometry* by J. Harris). This variety is called the *join* of V and W.

Let $A = [a_0 : \cdots : a_n]$ and $B = [b_0 : \cdots : b_n]$ be points of V and W respectively. Then any point $R = [z_0 : \cdots : z_n]$ of the line passing through P and Q is given by

$$\begin{pmatrix} z_0 \\ \vdots \\ z_n \end{pmatrix} = s \begin{pmatrix} a_0 \\ \vdots \\ a_n \end{pmatrix} + t \begin{pmatrix} b_0 \\ \vdots \\ b_n \end{pmatrix} \tag{1}$$

for some $[s : t] \in \mathbb{P}^1$. Let $(f_1, \ldots, f_l)$ and $(g_1, \ldots, g_m)$ be the generating sets for $I(V)$ and $I(W)$ respectively. Then the ideal defining $J(V,W)$ is obtained by solving the system of equations $f_1 = \cdots = f_l = 0$, $g_1 = \cdots = g_m = 0$ and (1) for $z_0, \ldots, z_{n-1}$ and z_n. By replacing x_i's by a_i' and b_i's, we obtained new ideals $I(V)_a$ in $k[a_0, \ldots, a_n]$ and $I(W)_b$ in $k[b_0, \ldots, b_n]$ respectively. Let I be the ideal in the following new ring:

$$k[a_0, \ldots, a_n, b_0, \ldots, b_n, s, t, z_0, \ldots, z_n]$$

generated by the equations in (1), and let $J = I + I(V)_a + I(W)_b$. Saturate J with respect to the ideal (s,t):

$$J' = J : (s,t)^\infty,$$

because we do not want s and t to vanish at the same time (recall that $[s : t] \in \mathbb{P}^1$). Take the intersection $\bar{J} = J' \cap k[z_0, \ldots, z_n]$. Then $V(\bar{J}) = J(V,W)$.

Let's make a `Macaulay2` script to compute the ideal of the join for two given varieties! The only tricky part is to replace variables. There we use a `while` loop:

```
i1 : idealOfJoin=(idl,idl')->(
          i:=0;
          R:=ring idl;
          KK:=coefficientRing R;
          n:=numgens(source vars R)-1;
          va:=vars ring idl;
          ringPn:=KK[a_0..a_n];
          avar:=vars ringPn;
          ringPn':=KK[b_0..b_n];
          bvar:=vars ringPn';
          S:=KK[a_0..a_n,b_0..b_n,s,t,va_(0,0)..va_(0,n),
          Degrees=>{2*n+4:1,n+1:2},MonomialOrder=>Eliminate (2*n+4)];
          idl=substitute(idl,S);
          idl'=substitute(idl',S);
          while i<n+1 do (
            idl=substitute(idl,{substitute(va_(0,i),S)
              =>(substitute(avar,S))_(0,i)});
            idl'=substitute(idl',{substitute(va_(0,i),S)
              =>(substitute(bvar,S))_(0,i)});
            i=i+1);
          gr:=ideal(substitute(vars R,S)-s*substitute(avar,S)
          -t*substitute(bvar,S))+idl+idl';
          sgr:=saturate(gr,ideal(s,t));
          elim:=ideal selectInSubring(1,gens gb sgr);
          joins:=substitute(elim,R);
          joins)

o1 = idealOfJoin

o1 : Function
```

Do you think that the program works? Let's check a few examples.

Example 1. Let P_1 and P_2 be two points in $\mathbb{P}^2$ chosen at random (so we can expect that P_1 and P_2 are distinct). What is $J(P_1, P_2)$? Obviously, $J(P_1, P_2)$ is the line passing through these two points:

```
i2 : KK=QQ;

i3 : ringP2=KK[x_0..x_2];

i4 : P1=ideal random(ringP2^{0},ringP2^{2:-1});
```

```
o4 : Ideal of ringP2

i5 : P2=ideal random(ringP2^{0},ringP2^{2:-1});

o5 : Ideal of ringP2

i6 : J=idealOfJoin(P1,P2)

                 69
o6 = ideal(x   - --*x )
            1    40  2

o6 : Ideal of ringP2
```

This J should define the line passing through P_1 and P_2. The line passing through these two points is defined by the linear form in $I(P_1) \cap I(P_2)$.

```
i7 : L=ideal(gens intersect(P1,P2))_{0}

                 69
o7 = ideal(x   - --*x )
            1    40  2

o7 : Ideal of ringP2

i8 : J==L

o8 = true
```

Example 2. Let L_1 and L_2 be two skew lines in $\mathbb{P}^4$. These two lines are contained in a hyperplane H, which is isomorphic to $\mathbb{P}^3$. So a line joining L_1 to L_2 intersects H in two points. This implies that this line lies in H (why?). Thus all we have to do is to determine $J(L_1, L_2)$ in H. The expected dimension of $J(L_1, L_2)$ is three, so we can expect that the lines joining L_1 to L_2 fill up H. Let's check this!

```
i9 : ringP4=KK[x_0..x_4];

i10 : L1=ideal random(ringP4^{0},ringP4^{3:-1});

o10 : Ideal of ringP4
```

```
i11 : L2=ideal random(ringP4^{0},ringP4^{3:-1});

o11 : Ideal of ringP4

i12 : J=idealOfJoin(L1,L2);

o12 : Ideal of ringP4

i13 : H=ideal(gens intersect(L1,L2))_{0};

o13 : Ideal of ringP4

i14 : J==H

o14 = true
```

Here `J` is the ideal of $J(L_1, L_2)$, and `H` is the hyperplane containing L_1 and L_2. `Macaulay2` tells us that these two ideals are the same, as we expected.

Remark 3. For each $i = 1, 2$, let $C(L_i)$ denote the set of lines hitting L_i. Recall that $C(L_1)$ and $C(L_2)$ are subvarieties in $\mathbb{G}(1, 3)$, where $\mathbb{G}(1, 3)$ is the grassmannian of lines in $\mathbb{P}^3$. More precisely, $C(L_1)$ and $C(L_2)$ correspond to hyperplane sections of $\mathbb{G}(1, 3)$. In other words, these are complete intersections of hyperplanes and $\mathbb{G}(1, 3)$ in $\mathbb{P}^5$ (recall that $\mathbb{G}(1, 3)$ is a quadric hypersurface in $\mathbb{P}^5$). So $Q = C(L_1) \cap C(L_2)$ is cut out by two hyperplanes and a quadric hypersurface, which means that Q is a quadric surface in $\mathbb{P}^3$. This quadric surface is expected to be smooth by Bertini's theorem (see page 216 in *Algebraic Geometry* by J. Harris for this theorem). Do you think that Q has something to do with the previous example? Any smooth quadric surface in $\mathbb{P}^3$ can be regarded as a surface obtained from $\mathbb{P}^1 \times \mathbb{P}^1$ by the Segre embedding. In our case, Q is $L_1 \times L_2$ embedded into H by the Segre embedding.

2. Secant varieties

Let V be a projective variety in $\mathbb{P}^n$. A line L is called a *secant line* to V if L meets V in two or more points. Let $I(V)$ be the ideal of V and let $(f_1, \ldots, f_l)$ be a generating set of $I(V)$. Given two points $A = [a_0 : \cdots : a_n]$ and $B = [b_0 : \cdots : b_n]$ of V, the line passing through A and B is expressed as (1). So we can hope that the set of secant lines to V can be described with the same spirit as in the case of joins. A problem arises, however, when A

and B coincide. Indeed, if $A = B$, then (1) gives rise to a point, which means that if we solve the system of equations (1), $f_1(A) = \cdots = f_l(A) = 0$ and $f_1(B) = \cdots = f_l(B) = 0$ for z_i's, then the set of secant lines does not fill up the solution set. However the solution set coincides with the closure of the set of secant lines. This closure is called the *secant variety* to V. We denote this variety by $\mathrm{Sec}(V)$. The ideal of $\mathrm{Sec}(V)$ can be computed by using the function `idealOfJoin`.

Exercise 1. Make a function for computing the ideal of the secant variety to a variety by using `idealOfJoin`. Using your `Macaulay2` script, compute the ideal of the secant variety to the Veronese surface in $\mathbb{P}^5$.

Let V be a projective variety of dimension k. Then the expected dimension of $\mathrm{Sec}(V)$ is $2k+1$ (why?). So the secant variety of the Veronese surface X in $\mathbb{P}^5$ is expected to have dimension 5, in other words, $\mathrm{Sec}(X) = \mathbb{P}^5$. However, the ideal of $\mathrm{Sec}(X)$ is generated by a single polynomial of degree 3! This implies that $\dim(\mathrm{Sec}(X)) = 4$.

Exercise 2. Explain why $\dim(\mathrm{Sec}(X)) = 4$.

Hint. Recall that the Veronese surface is obtained as the image of the map φ from $\mathbb{P}^2$ to $\mathbb{P}^5$ defined by

$$\varphi([x_0 : x_1 : x_2]) = [x_0^2 : x_0x_1 : x_1^2 : x_0x_2 : x_1x_2 : x_2^3].$$

We have proved that the map is an isomorphism. Note that each line in $\mathbb{P}^2$ is mapped to a plane conic on X. So, for any pair of points on X, there exists a unique conic on X. On the other hand, the family of conics on X is identified with the set of lines in $\mathbb{P}^2$, i.e. with $(\mathbb{P}^2)^*$. Use the fact that each conic spans a plane in $\mathbb{P}^5$.